Geodesics in Schwarzschild metric

using Black Hole Perturbation Toolkit

Md Arif Shaikh, Postdoc, ICTS-TIFR

March 19, 2021

Load the metric

There are built-in metric for most of the well known metrics but one can also manually set any metric. For working in Schwarzschild metric we need to do the following

In[1]:=
```
<< GeneralRelativityTensors`
```

In[2]:=
```
g = ToMetric["Schwarzschild"]
```

Out[2]= $g_{\alpha\beta}$

Then one can view the metric elements using TensorValues

In[3]:=
```
g // TensorValues // MatrixForm
```

Out[3]//MatrixForm=

$$\begin{pmatrix} -1+\frac{2M}{r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2M}{r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\mathrm{Sin}[\theta]^2 \end{pmatrix}$$

And access the metric elements

In[4]:=
```
g[-t, -t]
```

Out[4]= $-1+\frac{2M}{r}$

Metric can also be set in the following way

In[5]:=
```
k = ToMetric[{"my-metric", "k"}, {t, r, θ, ϕ},
  DiagonalMatrix[{-(1 - 2M/r), 1/(1 - 2M/r), r^2, r^2 Sin[θ]^2}], "Greek"]
```

Out[5]= $k_{\alpha\beta}$

In[6]:=
```
k // TensorValues // MatrixForm
```

Out[6]//MatrixForm=

$$\begin{pmatrix} -1+\frac{2M}{r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2M}{r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\mathrm{Sin}[\theta]^2 \end{pmatrix}$$

We get the same metric as the built-in one.

Christoffel Symbols, Riemann Tensor, Ricci Tensor and Ricci Scalar

Once we have our metric we can easily compute the other qualities using built-in functions

In[7]:=
```
Gammas = ChristoffelSymbol[g, ActWith → Simplify]
```

Out[7]= $\Gamma^{\alpha}{}_{\beta\gamma}$

In[8]:=
```
Gammas // TensorValues
```

Out[8]=

$$\left\{\left\{\left\{0,-\frac{M}{2Mr-r^2},0,0\right\},\left\{-\frac{M}{2Mr-r^2},0,0,0\right\},\{0,0,0,0\},\{0,0,0,0\}\right\},\right.$$
$$\left\{\left\{\frac{M(-2M+r)}{r^3},0,0,0\right\},\left\{0,\frac{M}{2Mr-r^2},0,0\right\},\right.$$
$$\left.\{0,0,2M-r,0\},\left\{0,0,0,(2M-r)\,\mathrm{Sin}[\theta]^2\right\}\right\},$$
$$\left\{\{0,0,0,0\},\left\{0,0,\frac{1}{r},0\right\},\left\{0,\frac{1}{r},0,0\right\},\{0,0,0,-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\}\right\},$$
$$\left.\left\{\{0,0,0,0\},\left\{0,0,0,\frac{1}{r}\right\},\{0,0,0,\mathrm{Cot}[\theta]\},\left\{0,\frac{1}{r},\mathrm{Cot}[\theta],0\right\}\right\}\right\}$$

In[9]:=
```
Gammas[t, t, r]
```

Out[9]=

$$-\frac{M\left(1-\frac{2M}{r}\right)r}{(2M-r)\left(2Mr-r^2\right)}$$

In[10]:=
```
Riemann = RiemannTensor[g, ActWith → Simplify]
```

Out[10]= $R_{\alpha\beta\gamma\delta}$

In[11]:=
```
Riemann // TensorValues
```

Out[11]=

$$\Big\{\Big\{\{\{0,0,0,0\},\{0,0,0,0\},\{0,0,0,0\},\{0,0,0,0\}\},$$
$$\Big\{\Big\{0,-\frac{2M}{r^3},0,0\Big\},\Big\{\frac{2M}{r^3},0,0,0\Big\},\{0,0,0,0\},\{0,0,0,0\}\Big\},$$
$$\Big\{\Big\{0,0,\frac{M(-2M+r)}{r^2},0\Big\},\{0,0,0,0\},\Big\{-\frac{M(-2M+r)}{r^2},0,0,0\Big\},\{0,0,0,0\}\Big\},$$
$$\Big\{\Big\{0,0,0,-\frac{M(2M-r)\,\mathrm{Sin}[\theta]^2}{r^2}\Big\},\{0,0,0,0\},$$
$$\{0,0,0,0\},\Big\{\frac{M(2M-r)\,\mathrm{Sin}[\theta]^2}{r^2},0,0,0\Big\}\Big\}\Big\},$$
$$\Big\{\Big\{\Big\{0,\frac{2M}{r^3},0,0\Big\},\Big\{-\frac{2M}{r^3},0,0,0\Big\},\{0,0,0,0\},\{0,0,0,0\}\Big\},$$
$$\{\{0,0,0,0\},\{0,0,0,0\},\{0,0,0,0\},\{0,0,0,0\}\},$$
$$\Big\{\{0,0,0,0\},\Big\{0,0,\frac{M}{2M-r},0\Big\},\Big\{0,-\frac{M}{2M-r},0,0\Big\},\{0,0,0,0\}\Big\},$$
$$\Big\{\{0,0,0,0\},\Big\{0,0,0,\frac{M\,\mathrm{Sin}[\theta]^2}{2M-r}\Big\},\{0,0,0,0\},\Big\{0,-\frac{M\,\mathrm{Sin}[\theta]^2}{2M-r},0,0\Big\}\Big\}\Big\},$$
$$\Big\{\Big\{\Big\{0,0,-\frac{M(-2M+r)}{r^2},0\Big\},\{0,0,0,0\},\Big\{\frac{M(-2M+r)}{r^2},0,0,0\Big\},\{0,0,0,0\}\Big\},$$
$$\Big\{\{0,0,0,0\},\Big\{0,0,-\frac{M}{2M-r},0\Big\},\Big\{0,\frac{M}{2M-r},0,0\Big\},\{0,0,0,0\}\Big\},$$
$$\{\{0,0,0,0\},\{0,0,0,0\},\{0,0,0,0\},\{0,0,0,0\}\},$$
$$\Big\{\{0,0,0,0\},\{0,0,0,0\},\Big\{0,0,0,2Mr\,\mathrm{Sin}[\theta]^2\Big\},\Big\{0,0,-2Mr\,\mathrm{Sin}[\theta]^2,0\Big\}\Big\}\Big\},$$
$$\Big\{\Big\{\Big\{0,0,0,\frac{M(2M-r)\,\mathrm{Sin}[\theta]^2}{r^2}\Big\},\{0,0,0,0\},$$
$$\{0,0,0,0\},\Big\{-\frac{M(2M-r)\,\mathrm{Sin}[\theta]^2}{r^2},0,0,0\Big\}\Big\},$$
$$\Big\{\{0,0,0,0\},\Big\{0,0,0,-\frac{M\,\mathrm{Sin}[\theta]^2}{2M-r}\Big\},\{0,0,0,0\},\Big\{0,\frac{M\,\mathrm{Sin}[\theta]^2}{2M-r},0,0\Big\}\Big\},$$
$$\Big\{\{0,0,0,0\},\{0,0,0,0\},\Big\{0,0,0,-2Mr\,\mathrm{Sin}[\theta]^2\Big\},\Big\{0,0,2Mr\,\mathrm{Sin}[\theta]^2,0\Big\}\Big\},$$
$$\{\{0,0,0,0\},\{0,0,0,0\},\{0,0,0,0\},\{0,0,0,0\}\}\Big\}\Big\}$$

In[12]:=
```
Riemann[t, -r, -t, -r]
```

Out[12]=

$$-\frac{2M}{(2M-r)\,r^2}$$

In[13]:=
```
Ricci = RicciTensor[g, ActWith → Simplify]
```
Out[13]= $R_{\beta\gamma}$

In[14]:=
```
Ricci // TensorValues
```
Out[14]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

In[15]:=
```
RS = RicciScalar[g, ActWith → Simplify]
```
Out[15]= R

In[16]:=
```
RS // TensorValues
```
Out[16]= 0

Set up Geodesic Equation using Christoffel Symbols

We can now set up the Geodesic Equations. First we set the four-velocity tensor

In[17]:=
```
u = ToTensor["four-velocity", g, {vt[τ], vr[τ], vθ[τ], vϕ[τ]}]
```
Out[17]= $\text{four}-\text{velocity}^{\alpha}$

In[18]:=
```
geodesicEq = D[TensorValues[u[α]], τ] +
  TensorValues[ContractIndices[Gammas[α, -μ, -ν] u[μ] u[ν]]]
```
Out[18]=

$$\left\{-\frac{2\,M\,vr[\tau]\,vt[\tau]}{2\,M\,r-r^2}+vt'[\tau],\right.$$
$$\frac{M\,vr[\tau]^2}{2\,M\,r-r^2}+\frac{M\,(-2\,M+r)\,vt[\tau]^2}{r^3}+(2\,M-r)\,v\theta[\tau]^2+(2\,M-r)\,\mathrm{Sin}[\theta]^2\,v\phi[\tau]^2+vr'[\tau],$$
$$\frac{2\,vr[\tau]\,v\theta[\tau]}{r}-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,v\phi[\tau]^2+v\theta'[\tau],$$
$$\left.\frac{2\,vr[\tau]\,v\phi[\tau]}{r}+2\,\mathrm{Cot}[\theta]\,v\theta[\tau]\,v\phi[\tau]+v\phi'[\tau]\right\}$$

t-component is then given by

In[19]:=
```
geodesicEq[[1]]
```
Out[19]=

$$-\frac{2\,M\,vr[\tau]\,vt[\tau]}{2\,M\,r-r^2}+vt'[\tau]$$

r-component could be obtained using

In[20]:=
```
geodesicEq[[2]]
```

Out[20]=
$$\frac{M\,vr[\tau]^2}{2\,M\,r-r^2}+\frac{M\,(-2\,M+r)\,vt[\tau]^2}{r^3}+(2\,M-r)\,v\theta[\tau]^2+(2\,M-r)\,\mathrm{Sin}[\theta]^2\,v\phi[\tau]^2+vr'[\tau]$$

and so on

Radial equation using Conserved quantities

In practice we don’t need to start with setting up the geodesic equations to get the radial equation. In fact it is easier to start with conserved quantities, Energy and the Angular momentum and the normalization condition to set and solve the radial equation. We can get the conserved quantities using the two killing vectors

In[21]:=
```
ξt = ToTensor["t-killing", g, {1, 0, 0, 0}]
```

Out[21]=
$$t-\mathrm{killing}^{\alpha}$$

In[22]:=
```
ξϕ = ToTensor["phi-killing", g, {0, 0, 0, 1}]
```

Out[22]=
$$phi-\mathrm{killing}^{\alpha}$$

In[23]:=
```
ContractIndices[ξt[-μ] u[μ]] // TensorValues
```

Out[23]=
$$\left(-1+\frac{2\,M}{r}\right)\,vt[\tau]$$

In[24]:=
```
ContractIndices[ξϕ[-μ] u[μ]] // TensorValues
```

Out[24]=
$$r^2\,\mathrm{Sin}[\theta]^2\,v\phi[\tau]$$

In[25]:=
```
norm = ContractIndices[u[-μ] u[μ]] // TensorValues
```

Out[25]=
$$\frac{vr[\tau]^2}{1-\frac{2\,M}{r}}+\left(-1+\frac{2\,M}{r}\right)\,vt[\tau]^2+r^2\,v\theta[\tau]^2+r^2\,\mathrm{Sin}[\theta]^2\,v\phi[\tau]^2$$

From the θ component of the geodesic equation we get

In[26]:=
```
θEq = geodesicEq[[3]]
```

Out[26]=
$$\frac{2\,vr[\tau]\,v\theta[\tau]}{r}-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,v\phi[\tau]^2+v\theta'[\tau]$$

Let' s consider equatorial motion, i.e., $\theta=\frac{\pi}{2}$

In[27]:=
```
θEqEquatorial = θEq /. {θ → π/2}
```
Out[27]=
$$\frac{2\,\text{vr}[\tau]\,\text{v}\theta[\tau]}{r} + \text{v}\theta'[\tau]$$

Now if we take the initial vθ = 0, then we get

In[28]:=
```
θEqEquatorial /. {vθ[τ] → 0}
```
Out[28]=
$$\text{v}\theta'[\tau]$$

Which implies that the derivative of vθ also goes to zero and hence θ does not change and remains $\frac{\pi}{2}$.
We use this in the norm expression to get

In[29]:=
```
normEq = norm /. {θ → π/2, vθ[τ] → 0}
```
Out[29]=
$$\frac{\text{vr}[\tau]^2}{1-\frac{2M}{r}} + \left(-1+\frac{2M}{r}\right)\text{vt}[\tau]^2 + r^2\,\text{v}\phi[\tau]^2$$

We now replace the vϕ and vt using the conserved quantities

In[30]:=
```
normEqnew = normEq /. {vt[τ] → EE/(1 - 2M/r), vϕ[τ] → L/r^2}
```
Out[30]=
$$\frac{\text{EE}^2\left(-1+\frac{2M}{r}\right)}{\left(1-\frac{2M}{r}\right)^2} + \frac{L^2}{r^2} + \frac{\text{vr}[\tau]^2}{1-\frac{2M}{r}}$$

In[45]:=
```
normCond = (normEqnew + 1) /. {EE^2 → 2 Enew + 1} // Simplify
```
Out[45]=
$$\frac{L^2}{r^2} - \frac{2\,(M+\text{Enew}\,r)}{-2M+r} + \frac{r\,\text{vr}[\tau]^2}{-2M+r}$$

From the above Expression one can obtain the equation for vr in terms of the Energy and the angular momentum.

Effective potential

In[46]:=
```
Eng = Solve[normCond == 0, Enew] // Values // Flatten
```
Out[46]=
$$\left\{\frac{-2L^2M + L^2 r - 2Mr^2 + r^3\,\text{vr}[\tau]^2}{2r^3}\right\}$$

In[48]:=
```
Veff = Eng[[1]] - 1/2 vr[τ]^2 // Simplify
```
Out[48]=
$$\frac{-2\,M\,r^2 + L^2\,(-2\,M + r)}{2\,r^3}$$

Plot Effective Potential vs radius

In[60]:=
```
Veffnew = Veff /. {r → r M, L → λ M} // Simplify
```
Out[60]=
$$\frac{-2\,r^2 - 2\,\lambda^2 + r\,\lambda^2}{2\,r^3}$$

In[92]:=
```
Lvals = {5, 4.2, 2 Sqrt[3], 3}
Show[Table[LogLinearPlot[Veffnew /. {λ → Lvals[[idx]]}, {r, 0, 100}],
  {idx, 1, Length@Lvals}], PlotRange → {{0, 5}, {-0.2, 0.2}} ]
```
Out[92]=
$$\left\{5,\ 4.2,\ 2\sqrt{3},\ 3\right\}$$

Out[93]=


ISCO

The inner most stable circular orbit (ISCO) is located at a place where Veff has a point of inflection. Thus we find where the first derivative of Veff is zero and second derivative is also zero.

In[103]:=
```
Vdash = D[Veffnew, r] // Simplify
```
Out[103]=
$$\frac{r^2 + 3\,\lambda^2 - r\,\lambda^2}{r^4}$$

In[106]:=
```
Vddash = D[Vdash, r] // Simplify
```

Out[106]=

$$\frac{-2\,r^2 - 12\,\lambda^2 + 3\,r\,\lambda^2}{r^5}$$

In[107]:=
```
Solve[{Vdash == 0, Vddash == 0}, {r, λ}]
```

Out[107]=

$$\left\{\left\{r \to 6,\ \lambda \to -2\sqrt{3}\right\},\ \left\{r \to 6,\ \lambda \to 2\sqrt{3}\right\}\right\}$$